

Opty: optimistic concurrency control

Jordi Guitart

Adapted with permission from Johan Montelius (KTH)

November 21, 2012

Introduction

In this session, you will implement a transaction server using optimistic concurrency control. You will also learn how to implement a updatable data structure in Erlang that can be accessed by concurrent, possibly distributed, processes. Before you start, you should know how optimistic concurrency control with backward validation works (read paper 'On Optimistic Methods for Concurrency Control' by Kung and Robinson).

1 The architecture

The architecture consists of a *transaction server* process having access to a *store* and a *validator*. The *store* consists of a set of *entries*, each holding a value and a unique reference. The reference is updated in each write operation so we can validate that nothing has happened with the *entry* since we last read its value.

A *client* starts a transaction by creating a *transaction handler* that is given access to the *store* and the *validator*. The *transaction server* is thus not involved in the transaction; it is simply a process from which we can get access to the *store* and the *validator*.

1.1 The handler

The *transaction handler* will handle read and write requests from the *client* and will also close the transaction. In each read operation, the *handler* keeps track of the unique reference of the *entry* that is read. It also keeps the write operation in a local store so that the real *store* is not modified until we want to close the transaction.

When the transaction is closed, the *handler* sends the read and write sets to the *validator*.

1.2 The validator

The *validator* process will be able to tell if a value of an *entry* has been changed since the transaction read the entry. If none of the *entries* have

changed, the transaction can commit and the associated write operations are performed.

Since there is only one *validator* in the system, it is the only process that actually writes anything to the *store*.

2 The implementation

Since the implementation has some complexity, it is probably a good idea that you draw a diagram containing the different involved processes and their interaction via messages in order to fully understand how the system works.

Since data structures in Erlang are not mutable (we cannot change their values), we need to do a trick. The *store* is represented by a tuple of process identifiers. Each process is an *entry* and we can change its value by sending it write messages.

A *transaction handler* is given a copy of the tuple (i.e. the *store*) when created, but the processes that represent the *entries* are of course not copied. The *store* can thus be shared by several *transaction handlers*, all who can send read messages to the *entries*.

The *validator* will not need access to the whole *store* since it will be given the read and write sets from the *transaction handlers*. These sets will contain the process identifiers of the relevant *entries*.

2.1 An entry

A process that implements an *entry* should only have a value and a unique reference as its state. We will below call this reference “timestamp” but it has nothing to do with time, it’s simply a unique reference. The reference will be given to a reader of the value so that the *validator* later can determine if the value has been changed. We could do without the reference but it has its advantages.

```
-module(entry).  
-export([new/1]).  
  
new(Value) ->  
    spawn_link(fun() -> init(Value) end).  
  
init(Value) ->  
    entry(Value, make_ref()).
```

Note that we are using the primitive `spawn_link/1`. This is to ensure that if the creator of the *entry* dies, then the *entry* should also die.

Three messages should be handled by an *entry*:

- `{read, Ref, From}`: a read request tagged with a reference. We will return a message tagged with the reference so that the requester can identify the correct message. The reply will contain also the process identifier of the *entry*, the value, and the current timestamp.
- `{write, Value}`: update the current value of the *entry*, no reply needed. The timestamp of the *entry* will be updated.
- `{check, Ref, Readtime, From}`: check if the timestamp of the *entry* has changed since we read the value (at time `Readtime`). A reply, tagged with the reference, will tell the *validator* if the timestamp is still the same or if the client has to abort.
- `stop`: terminate.

```
entry(Value, Time) ->
  receive
    {read, Ref, From} ->
      %% TODO: ADD SOME CODE
      entry(Value, Time);
    {write, New} ->
      entry(New, make_ref());
    {check, Ref, Readtime, From} ->
      if
        Readtime == Time ->
          From ! {Ref, ok};
        true ->
          From ! {Ref, abort}
      end,
      entry(Value, Time);
    stop ->
      ok
  end.
```

Why do we want the reply to the read message to include the process identifier? Not quite clear at the moment but you will see that it makes it easy to write an asynchronous *transaction handler*.

2.2 The store

We will hide the representation of the *store* and provide only an API to create a new *store* consisting of *N entries* and to lookup an *entry* in the *store*. Creating a tuple is done simply by first creating a list of all process identifiers and then turning it into a tuple.

```

-module(store).
-export([new/1, stop/1, lookup/2]).

new(N) ->
    list_to_tuple(entries(N, [])).

stop(Store) ->
    lists:map(fun(E) ->
                E ! stop
            end,
              tuple_to_list(Store)).

lookup(I, Store) ->
    element(I, Store). % this is a builtin function

entries(N, Sofar) ->
    if
        N == 0 ->
            Sofar;
        true ->
            Entry = entry:new(0),
            entries(N-1, [Entry|Sofar])
    end.

```

2.3 The transaction handler

A *client* should never access the *store* directly. It will perform all operations through a *transaction handler*. A *transaction handler* is created for a specific *client* and holds the *store* and the process identifier of the *validator*.

We will implement the *handler* so that a *client* can make asynchronous reads to the *store*. If latencies are high, there is no point in waiting for one read operation to complete before initiating a second operation.

The task of the *transaction handler* is to record all read operations (and at what time these took place) and make write operations only visible in a local store. In order to achieve this, the *handler* will keep two sets: the read set (**Reads**) and the write set (**Writes**). The read set is a list of tuples {Entry, Time} and the write set is a list of tuples {N, Entry, Value}. When it is time to commit, the *handler* sends the read and write sets to the *validator*.

When the *handler* is created it is also linked to its creator. This means that if one dies both die. This sounds hard but it will be explained once we implement the *server*.

```

-module(handler).

```

```
-export([start/3]).
```

```
start(Client, Validator, Store) ->  
    spawn_link(fun() -> init(Client, Validator, Store) end).
```

```
init(Client, Validator, Store) ->  
    handler(Client, Validator, Store, [], []).
```

The message interface to the *handler* is as follows:

- `{read, Ref, N}`: a read request from the *client* containing a reference that we should use in the reply message. The integer *N* is the index of the *entry* in the *store*. The *handler* should first look through the write set to see if *entry* *N* has been written. In this case, the written value is returned to the *client*. If no matching operation is found, a message is sent to the *Nth entry* process in the *store*. This *entry* will reply to the *handler* since we need to record the read time.
- `{Ref, Entry, Value, Time}`: a reply from an *entry* that should be forwarded to the *client*. The `{Entry, Time}` tuple is saved in the read set of the *handler*. The reply to the *client* is `{value, Ref, Value}`.
- `{write, N, Value}`: a write request from the *client*. The integer *N* is the index of the *entry* in the *store* and *Value*, the new value. The `{N, Entry, Value}` tuple is saved in the write set of the *handler*.
- `{commit, Ref}`: a commit request from the *client*. This is the time to contact the *validator* and see if there are any conflicts in our read set. If not, the *validator* will perform the write operations in the write set and reply directly to the *client*.

Here is some skeleton code for the *handler*.

```
handler(Client, Validator, Store, Reads, Writes) ->  
    receive  
        {read, Ref, N} ->  
            case lists:keysearch(N, 1, Writes) of  
                {value, {N, _, Value}} ->  
                    %% TODO: ADD SOME CODE  
                    handler(Client, Validator, Store, Reads, Writes);  
                false ->  
                    %% TODO: ADD SOME CODE  
                    %% TODO: ADD SOME CODE  
                    handler(Client, Validator, Store, Reads, Writes)  
            end;  
        {Ref, Entry, Value, Time} ->
```

```

        %% TODO: ADD SOME CODE HERE AND COMPLETE NEXT LINE
        handler(Client, Validator, Store, [...|Reads], Writes);
    {write, N, Value} ->
        %% TODO: ADD SOME CODE HERE AND COMPLETE NEXT LINE
        Added = lists:keystore(N, 1, Writes, {N, ..., ...}),
        handler(Client, Validator, Store, Reads, Added);
    {commit, Ref} ->
        Validator ! {validate, Ref, Reads, Writes, Client};
    abort ->
        ok
end.

```

2.4 The validator

The *validator* is responsible of doing the final validation of transactions. The task is made quite easy since only one transaction is validated at a time. There are no concurrent operations that could possibly conflict with the validation process.

When we start the *validator*, we also link it to the process that creates it. This is to ensure that we don't have any zombie process.

```

-module(validator).
-export([start/0]).

start() ->
    spawn_link(fun() -> init() end).

init()->
    validator().

```

The *validator* receives a request from a *handler* containing everything that is needed both to validate that the transaction is allowed and to perform the write operations that will be a result of the transaction. The request contains:

- **Ref:** a unique reference to tag the reply message.
- **Reads:** the list of read operations that have been performed. The *validator* must ensure that the *entries* of the read operations have not been changed.
- **Writes:** the pending write operations that, if the transaction is valid, should be applied to the *store*.
- **Client:** the process identifier of the *client* to whom we should return the reply.

Validation is thus simply checking if read operations are still valid and if so update the *store* with the pending write operations.

```

validator() ->
  receive
    {validate, Ref, Reads, Writes, Client} ->
      case validate(Reads) of
        ok ->
          update(Writes),
          Client ! {Ref, ok};
        abort ->
          Client ! {Ref, abort}
      end,
      validator();
    _Old ->
      validator()
  end.

update(Writes) ->
  lists:map(fun({_ , Entry, Value}) ->
    %% TODO: ADD SOME CODE
  end,
  Writes).

```

Since a read operation is represented with a tuple {*Entry*, *Time*}, the *validator* needs only to send a **check** message to the *entry* and make sure that the current timestamp of the *entry* is the same.

```

validate(Reads) ->
  {N, Tag} = send_checks(Reads),
  check_reads(N, Tag).

```

For better performance the *validator* can first send check messages to all *entries* and then collect the replies. As soon as one *entry* replies with an **abort** message, we're done. Note however, that we must be careful so that we are not seeing replies that pertain to a previous validation. When we send our check request we therefore tag replies with a unique reference so that we know that we're counting the right replies.

```

send_checks(Reads) ->
  Tag = make_ref(),
  Self = self(),
  N = length(Reads),
  lists:map(fun({Entry, Time}) ->
    Entry ! {check, Tag, Time, Self}
  end,
  Reads).

```

```

        end,
        Reads),
    {N, Tag}.

```

Collecting the replies is a simple task and we only have to be careful so that we don't collect an old reply.

```

check_reads(N, Tag) ->
    if
        N == 0 ->
            ok;
        true ->
            receive
                {Tag, ok} ->
                    check_reads(N-1, Tag);
                {Tag, abort} ->
                    abort
            end
    end.

```

Old messages that are still in the queue must be removed somehow, this is why, and this is very important, the main loop of the *validator* includes a catch all clause.

2.5 The server and the client

We now have all the pieces to build the *transaction server*. The *server* will construct the *store* and the *validator* process. *Clients* can then open a transaction and each transaction will be given a new *transaction handler* that is dedicated to the task.

```

-module(server).
-export([start/1]).

start(N) ->
    spawn(fun() -> init(N) end).

init(N) ->
    Store = store:new(N),
    Validator = validator:start(),
    server(Validator, Store).

server(Validator, Store) ->
    receive
        {open, Client} ->

```



```

        %% TODO: ADD SOME CODE
        server(Validator, Store);
    stop ->
        store:stop(Store)
end.

```

The *server* could simply wait for requests from *clients* and spawn a new *transaction handler*. There is however a trap here that we don't want to fall into. If the *server* creates the *transaction handler*, we must let the *handler* be independent from the *server* (not linked), because if the *handler* dies we don't want the *server* to die. On the other hand, if the *client* dies we do want the *handler* to die. The solution is to let the process of the *client* create the *handler* when it receives the info about the *validator* and the *store* from the *server*. Then, it moves to a loop where it executes transactions until a *stop* message is received.

```

-module(client).
-export([start/4]).

```

```

start(Name, Entries, Updates, Server) ->
    spawn(fun() -> open(Name, Entries, Updates, Server, 0, 0) end).

```

```

open(Name, Entries, Updates, Server, Total, Ok) ->
    {A1,A2,A3} = now(),
    random:seed(A1, A2, A3),
    Server ! {open, self()},
    receive
        {stop, From} ->
            io:format("~w: Transactions TOTAL:~w, OK:~w, -> ~w % ~n",
                [Name, Total, Ok, 100*Ok/Total]),
            From ! {done, self()},
            ok;
        {transaction, Validator, Store} ->
            Handler = handler:start(self(), Validator, Store),
            do_transactions(Name, Entries, Updates, Server, Handler,
                Total, Ok, Updates)
    end.

```

```

% Commit transaction
do_transactions(Name, Entries, Updates, Server, Handler, Total, Ok, 0) ->
    %io:format("~w: Commit: TOTAL ~w, OK ~w~n", [Name, Total, Ok]),
    %timer:sleep(Name*10),
    Ref = make_ref(),
    Handler ! {commit, Ref},

```

```

Result = receiveCommitValue(Ref),
if
    Result == ok ->
        open(Name, Entries, Updates, Server, Total+1, Ok+1);
    true ->
        open(Name, Entries, Updates, Server, Total+1, Ok)
end;

% Reads and Writes
do_transactions(Name, Entries, Updates, Server, Handler, Total, Ok, N) ->
    %io:format("~w: R/W: TOTAL ~w, OK ~w, N ~w~n", [Name, Total, Ok, N]),
    Ref = make_ref(),
    Num = random:uniform(Entries),
    Handler ! {read, Ref, Num},
    Value = receiveValue(Ref),
    Handler ! {write, Num, Value+1},
    do_transactions(Name, Entries, Updates, Server, Handler, Total, Ok, N-1).

receiveCommitValue(Ref) ->
    receive
        {Ref, Value} -> Value
    end.

receiveValue(Ref) ->
    receive
        {value, Ref, Value} -> Value
    end.

-module(opty).
-export([start/4, stop/1]).

start(Clients, Entries, Updates, Time) ->
    register(s, server:start(Entries)),
    L = startClients(Clients, [], Entries, Updates),
    io:format("Starting: ~w CLIENTS, ~w ENTRIES, ~w UPDATES PER TRANSACTION,
    DURATION ~w s ~n", [Clients, Entries, Updates, Time]),
    timer:sleep(Time*1000),
    stop(L).

stop(L) ->
    io:format("Stopping...~n"),
    stopClients(L),
    s ! stop.

```

```

startClients(0,L,_,_) -> L;
startClients(Clients, L, Entries, Updates) ->
    Pid = client:start(Clients, Entries, Updates, s),
    startClients(Clients-1, [Pid|L], Entries, Updates).

stopClients([]) -> ok;
stopClients([Pid|L]) ->
    Pid ! {stop, self()},
    receive
        {done, Pid} -> ok
    end,
    stopClients(L).

```

You're done, you have implemented a transaction server using optimistic concurrency control.

3 Performance

Open Questions. What is the impact of the number of concurrent transactions on the success rate (i.e. percentage of committed transactions with respect to the total)? This of course depends on the size of the store, how many write operations each transaction does, and how much delay we have in between the read instructions of the transaction and the final commit instruction.

According to this, to assess the performance of your transaction server, you might need to create a benchmark (updating the client provided if needed) which creates several clients executing several concurrent transactions allowing you to test: (i) different number of entries in the store, (ii) different duration of the transactions – e.g., each transaction takes around 1 second to commit (you might use a sleep command to artificially delay the commit operation), (iii) different ratio of read and write operations per transaction, (iv) different percentage of modified entries with respect to the total number of entries – for example, each transaction modifies randomly 20% of entries. Thus, you will be able to play with these parameters and see when conflicts arise.

Split the `opty` module in two parts and make the needed adaptations to enable the server and the clients to run in different machines. Remember how names registered in remote nodes are referred and how Erlang runtime should be started to run distributed programs.

If we run this in a distributed Erlang network, where is the handler running? What is actually copied when a transaction handler is started?